

**tsline** — Plot time-series data

[Description](#)  
[Options](#)

[Quick start](#)  
[Remarks and examples](#)

[Menu](#)  
[References](#)

[Syntax](#)  
[Also see](#)

## Description

`tsline` draws line plots for time-series data.

`tsrline` draws a range plot with lines for time-series data.

## Quick start

Line plot for the time series `y1` using `tsset` data

```
tsline y1
```

Add plots of time series `y2` and `y3`

```
tsline y1 y2 y3
```

Range plot with lines for the lower and upper values of time series `y1` stored in `y1_lower` and `y1_upper`, respectively

```
tsrline y1_lower y1_upper
```

Overlay a range plot of the lower and upper values of time series `y1` stored in `y1_lower` and `y1_upper`, respectively, on a plot of `y1`

```
tsline y1 || tsrline y1_lower y1_upper
```

## Menu

Statistics > Time series > Graphs > Line plots

## Syntax

*Time-series line plot*

```
[twoway] tsline varlist [if] [in] [, tsline_options]
```

*Time-series range plot with lines*

```
[twoway] tsrline y1 y2 [if] [in] [, tsrline_options]
```

where the time variable is assumed set by `tsset` (see [TS] **tsset**), *varlist* has the interpretation  $y_1 [y_2 \dots y_k]$ .

<i>tsline_options</i>	Description
Plots	
<i>scatter_options</i>	any options documented in [G-2] <b>graph twoway scatter</b> with the exception of <i>marker_options</i> , <i>marker_placement_options</i> , and <i>marker_label_options</i> , which will be ignored if specified
Y axis, Time axis, Titles, Legend, Overall, By	
<i>twoway_options</i>	any options documented in [G-3] <i>twoway_options</i>

<i>tsrline_options</i>	Description
Plots	
<i>rline_options</i>	any options documented in [G-2] <b>graph twoway rline</b>
Y axis, Time axis, Titles, Legend, Overall, By	
<i>twoway_options</i>	any options documented in [G-3] <i>twoway_options</i>

## Options

### Plots

*scatter\_options* are any of the options allowed by the `graph twoway scatter` command except that *marker\_options*, *marker\_placement\_option*, and *marker\_label\_options* will be ignored if specified; see [G-2] **graph twoway scatter**.

*rline\_options* are any of the options allowed by the `graph twoway rline` command; see [G-2] **graph twoway rline**.

### Y axis, Time axis, Titles, Legend, Overall, By

*twoway\_options* are any of the options documented in [G-3] *twoway\_options*. These include options for titling the graph (see [G-3] *title\_options*), options for saving the graph to disk (see [G-3] *saving\_option*), and the `by()` option, which will allow you to simultaneously plot different subsets of the data (see [G-3] *by\_option*).

Also see the `recast()` option discussed in [G-3] *advanced\_options* for information on how to plot spikes, bars, etc., instead of lines.

## Remarks and examples

Remarks are presented under the following headings:


[Basic examples](#)  
[Advanced example](#)  
[Video example](#)

### Basic examples

#### ► Example 1: A time-series line plot

We simulated two separate time series (each of 200 observations) and placed them in a Stata dataset, `tsline1.dta`. The first series simulates an AR(2) process with  $\phi_1 = 0.8$  and  $\phi_2 = 0.2$ ; the second series simulates an MA(2) process with  $\theta_1 = 0.8$  and  $\theta_2 = 0.2$ . We use `tsline` to graph these two series.

```
. use http://www.stata-press.com/data/r14/tsline1
. tsset lags
    time variable: lags, 0 to 199
                  delta: 1 unit
. tsline ar ma
```




◀

#### ► Example 2: Using options to highlight information

Suppose that we kept a calorie log for an entire calendar year. At the end of the year, we would have a dataset (for example, `tsline2.dta`) that contains the number of calories consumed for 365 days. We could then use `tsset` to identify the date variable and `tsline` to plot calories versus time. Knowing that we tend to eat a little more food on Thanksgiving and Christmas day, we use the `ttick()` and `tttext()` options to point out these days on the time axis.

```
. use http://www.stata-press.com/data/r14/tsline2
. tsset day
    time variable:  day, 01jan2002 to 31dec2002
      delta: 1 day
. tsline calories, ttick(28nov2002 25dec2002, tpos(in))
> ttext(3470 28nov2002 "thanks" 3470 25dec2002 "x-mas", orient(vert))
```




Options associated with the time axis allow dates (and times) to be specified in place of numeric date (and time) values. For instance, we used

```
ttick(28nov2002 25dec2002, tpos(in))
```

to place tick marks at the specified dates. This works similarly for `tlabel()`, `tmlabel()`, and `tmtick()`.

Suppose that we wanted to place vertical lines for the previously mentioned holidays. We could specify the dates in the `tline()` option as follows:


```
. tsline calories, tline(28nov2002 25dec2002)
```



### ▷ Example 3: Formatting the time axis

We could also modify the format of the time axis so that the labeled ticks display only the day in the year:

```
. tsline calories, tlabel(, format(%tdmd)) ttitle("Date (2002)")
```



◀

### Advanced example

`tsline` and `tsrline` are both commands and *plottypes* as defined in [G-2] **graph twoway**. Thus the syntax for `tsline` is

```
. graph twoway tsline ...
. twoway tsline ...
. tsline ...
```

and similarly for `tsrline`. Being plot types, these commands may be combined with other plot types in the `twoway` family, as in,

```
. twoway (tsrline ...) (tsline ...) (lfit ...) ...
```


which can equivalently be written as

```
. tsrline ... || tsline ... || lfit ... || ...
```

## ▷ Example 4: Combining line and range plots

In the first plot of [example 2](#), we were uncertain of the exact values we logged, so we also gave a range for each day. Here is a plot of the summer months.

```
. tsrline lcalories ucalories if tin(1may2002,31aug2002) || tsline calories ||
> if tin(1may2002,31aug2002), ytitle(Calories)
```



4

## Video example

[Time series, part 2: Line graphs and tin\(\)](#)

## References

- Cox, N. J. 2009a. [Speaking Stata: Graphs for all seasons](#). *Stata Journal* 6: 397–419.
- . 2009b. [Stata tip 76: Separating seasonal time series](#). *Stata Journal* 9: 321–326.
- . 2012. [Speaking Stata: Transforming the time axis](#). *Stata Journal* 12: 332–341.

## Also see

- [TS] [tsset](#) — Declare data to be time-series data
- [G-2] [graph twoway](#) — Twoway graphs
- [XT] [xtline](#) — Panel-data line plots